

The Uses of Mixed Models

Introductory Examples (from Applied Mixed Models in Medicine)

Example 1: Utilization of incomplete information in a cross-over trial.

Cross-over design. Each animal/patient receives all treatments and response to each treatment is measured.

- used to assess treatment efficacy in “chronic” conditions
- gives benefit of ‘within animal/patient’ comparison

Patient	Treatment		Differences A-B	Patient Mean
1	20	12	8	16.0
2	26	24	2	25.0
3	16	17	-1	16.5
4	29	21	8	25.0
5	22	21	1	21.5
6	24	17	7	20.5
Mean	22.83	18.67	4.17	20.75

SAS CODE:

```
DATA CROSSOVER;  
  INPUT PAT TRT $ Y;  
CARDS;  
1 A 20  
1 B 12  
2 A 26  
2 B 24  
3 A 16  
3 B 17  
4 A 29  
4 B 21  
5 A 22  
5 B 21  
6 A 24  
6 B 17  
;  
RUN;
```

Model A – Assess Treatment effects only

$$y_{ij} = \mu + \tau_i + e_{ij}$$

Model assumptions

$$e_{ij} \sim N(0, \sigma^2)$$

$$\text{var}(y_{ij}) = \sigma^2$$

$$\text{cov}(y_{ij}, y_{i'j'}) = \text{cov}(\mu + \tau_j + e_{ij}, \mu + \tau_{j'} + e_{i'j'}) = \text{cov}(e_{i'j'}, e_{ij}) = 0$$

20.75 = estimate of μ

22.83 = estimate of $\mu + \tau_A$

22.83 - 20.75 = 2.08 = estimate of τ_A

Similarly 18.67 - 20.75 = -2.08 = estimate of τ_B

SAS CODE:

```
PROC GLM;
CLASS TRT;
MODEL Y=TRT;
LSMEANS TRT/PDIFF;
RUN;
```

Source	DF	Sum of Squares	Mean Square		
Model	1	52.0833333	52.0833333		
Error	10	194.1666667	19.4166667		

Source	DF	SS	Mean Square	F Value	Pr > F
trt	1	52.08333333	52.08333333	2.68	0.1325

ANOVA table

Source of Variation	Degrees of Freedom	Sums of Squares	Mean Square	F	p
Treatment	1	52.08	52.08	2.68	0.13
Residual	10	194.17	19.42		

Estimated difference $2.08 - (-2.08) = 4.16$

$$SE(\tau_A - \tau_B) = \sqrt{\sigma^2(1/n_A + 1/n_B)} = \sqrt{(2 \times \sigma^2 / 6)} = \sqrt{6.47} = 2.54$$

t test

$$t=4.16/2.54=1.63$$

Model B – taking into account patient effect

$$y_{ij} = \mu + p_i + \tau_j + e_{ij}$$

SAS CODE:

```
PROC GLM;
CLASS PAT TRT;
MODEL Y=PAT TRT;
LSMEANS TRT/PDIFF;
RUN;
```

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	6	206.8333333	34.4722222	4.37	0.0634
Error	5	39.4166667	7.8833333		

Source	DF	SS	Mean Square	F Value	Pr > F
pat	5	154.7500000	30.9500000	3.93	0.0798
trt	1	52.0833333	52.0833333	6.61	0.0500

ANOVA table

Source of Variation	Degrees of Freedom	Sums of Squares	Mean Square	F	p
Patient	5	154.75	30.95	3.93	0.08
Treatment	1	52.08	52.08	6.61	0.05
Residual	5	39.42	7.88		

Residual represents “within-patient variation”.

$$SE(\tau_A - \tau_B) = \sqrt{\sigma^2(1/n_A + 1/n_B)} = \sqrt{(2 \times \sigma^2 / 6)} = \sqrt{2.63} = 1.62$$

Model C – random effects model

Only assumption made for Models A and B about variation was residuals normally distributed. Assumed patient and treatment effects took constant values or “fixed effects”. Models A and B are **fixed effects models**.

Assume patients are realizations of values from a probability distribution.
If patients independent samples from a normal distribution

$$y_{ij} = \mu + p_i + \tau_j + e_{ij}$$

$$e_{ij} \sim \mathbf{N}(0, \sigma^2)$$

$$p_i \sim \mathbf{N}(0, \sigma_p^2)$$

p_i is called a “random effect”

So the above model is considered a “mixed model”, which is a model that contains both fixed and random effects.

This class of models, which includes random effects will be referred to as “random effects” models to distinguish them from models that will be presented later.

Variance component

Model parameter that quantifies the random variation due to a random effect only.

σ_p^2 is the variance component for random patient effect in the model above.

$$\text{var}(y_{ij}) = \sigma_p^2 + \sigma^2$$

$$\text{cov}(y_{ij}, y_{i'j'}) = \text{cov}(\mu + p_i + \tau_j + e_{ij}, \mu + p_{i'} + \tau_{j'} + e_{i'j'}) = \text{cov}(p_i + e_{ij}, p_{i'} + e_{i'j'})$$

When patients are different $i \neq i'$, $\text{cov}(y_{ij}, y_{i'j'}) = 0$

When two samples are from the same patient then

$$\text{cov}(y_{ij}, y_{i'j'}) = \text{cov}(p_i, p_i) = \sigma_p^2$$

In other words, the observations on the same patient are correlated, while observations on different patients are uncorrelated.

SAS CODE:

```
PROC MIXED;
CLASS TRT PAT;
MODEL Y=TRT;
RANDOM PAT;
RUN;
```

The Mixed Procedure

Cov Parm	Estimate
pat	11.5333
Residual	7.8833

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
trt	1	5	6.61	0.0500

ANOVA table

Source of Variation	D. F.	S.S.	M.S.	E(M.S.)	F	p
Patient	5	154.75	30.95	$\sigma^2 + 2\sigma_p^2$	3.93	0.08
Treatment	1	52.08	52.08	$\sigma^2 + 6\sum\tau_i^2$	6.61	0.05
Residual	10	39.42	7.88	σ^2		

$$\hat{\sigma}^2 = 7.88$$

$$\hat{\sigma}^2 + 2\hat{\sigma}_p^2 = 30.95$$

$$7.88 + 2\hat{\sigma}_p^2 = 30.95$$

$$2\hat{\sigma}_p^2 = 30.95 - 7.88 = 23.07$$

$$\hat{\sigma}_p^2 = 11.54$$

With no missing data, the results from the ANOVA with regards to treatment were the same, whether patient was treated random or fixed. What happens if there are any data missing?

Data with missing values

Patient	Treatment		Differences A-B	Patient Mean
1	20	12	8	16.0
2	26	24	2	25.0
3	16	17	-1	16.5
4	29	21	8	25.0
5	22	-	-	22.0
6	-	17	-	17.0
Mean			4.25	

SAS CODE:

```
DATA CROSSOVER;
  INPUT PAT TRT $ Y;
CARDS;
1 A 20
1 B 12
2 A 26
2 B 24
3 A 16
3 B 17
4 A 29
4 B 21
5 A 22
6 B 17
;
RUN;
```

Model B – taking into account patient effect as fixed

SAS CODE:

```
PROC GLM;
  CLASS PAT TRT;
  MODEL Y=PAT TRT;
  LSMEANS TRT/PDIFF;
  ESTIMATE 'TREATMENT DIFF' TRT 1 -1;
  RUN;
```

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	6	204.0250000	34.0041667	3.36	0.1738
Error	3	30.3750000	10.1250000		

Source	DF	Type I SS	Mean Square	F Value	Pr > F
pat	5	167.9000000	33.5800000	3.32	0.1763
trt	1	36.1250000	36.1250000	3.57	0.1553

Standard

Parameter	Estimate	Error	t Value	Pr > t
treatment diff	4.25000000	2.25000000	1.89	0.1553

Note that the estimate of the treatment difference calculated only from data from patients 1-4.

$$\hat{\sigma}^2 = 10.125$$

$$SE(\tau_A - \tau_B) = \sqrt{\sigma^2(1/n_A + 1/n_B)} = \sqrt{(2 \times \sigma^2 / 4)} = \sqrt{5.06} = 2.25$$

$$t = 4.25 / 2.25 = 1.89$$

Model C – taking into account patient effect as random

SAS CODE:

```
PROC MIXED;  
CLASS TRT PAT;  
MODEL Y=TRT;  
RANDOM PAT;  
ESTIMATE 'TREATMENT DIFF' TRT 1 -1;  
RUN;
```

	Cov Parm	Estimate				
	pat	12.6271				
	Residual	8.8992				
Label	Estimate	Error	DF	t Value	Pr > t	
treatment diff	4.3203	2.0082	3	2.15	0.1206	

$$\hat{\sigma}^2 = 8.899$$

$$\hat{\sigma}_p^2 = 12.627$$

Note that the standard error for the treatment difference (2.008) is smaller than the standard error of the treatment difference obtained from the fixed effect model (2.25).

Due to:

- fortuitously lower estimate of σ^2
- random effects model uses information on treatment from both between patient error (σ_p^2) and within patient error (σ^2)

For the fixed effect model, the standard error of the estimate only uses the within patient error.

This extra information used in the random effects model compared to the fixed effects model is referred to as the *recovery* of between-patient information.

Example above shows how using a random effects model uses information that would have been lost in a fixed effect analysis.

Estimation (or prediction) of random effects

We assumed patients were a random sample drawn from a population.

$$p_i \sim N(0, \sigma_p^2)$$

Want to determine or “predict” the location within the normal distribution from which each patient’s observation came from.

Using the complete dataset, the following SAS code estimates the solutions deviated from the mean for all of the patients in the study.

SAS CODE:

```
PROC MIXED;  
CLASS TRT PAT;  
MODEL Y=TRT;  
RANDOM PAT/SOLUTION;
```

Solution for Random Effects

Effect	pat	Estimate	Std Err Pred	DF	t Value	Pr > t
pat	1	-3.5401	2.0905	5	-1.69	0.1512
pat	2	3.1675	2.0905	5	1.52	0.1902
pat	3	-3.1675	2.0905	5	-1.52	0.1902
pat	4	3.1675	2.0905	5	1.52	0.1902
pat	5	0.5590	2.0905	5	0.27	0.7999
pat	6	-0.1863	2.0905	5	-0.09	0.9324

The prediction for each patient is the sum of the mean plus the solution. Compare these solutions with the fixed effects model estimates.

Patient	Fixed	Random
1	16.0	-3.5401+20.75=17.2
2	25.0	3.1675+20.75=23.9
3	16.5	-3.1675+20.75=17.6
4	25.0	3.1675+20.75=23.9
5	21.5	0.5590+20.75=21.3
6	20.5	-0.1863+20.75=20.6

The predictions from the random effects model differ from the corresponding estimates from the fixed effects model. That is because the “prediction” for each patient is affected by the “predictions” for all other patients. These predictions are less widely spread and therefore are described as “shrunk”. This shrinkage occurs because patients are a sample from the overall patient population.

Extent of shrinkage depends on the relative sizes of σ_p^2 and σ^2 .

If $\sigma_p^2 = 0$, all patient predictions will be the same.

The fewer observations per patient, the greater σ^2 (within patient error) and therefor greater relative shrinkage.

Shrinkage occurs for both balanced and unbalanced data. We will see the formula later.

Example 2: A multi-center hypertension trial

The following are data that are used extensively in examples throughout the book.

Randomized, double blind comparison of 3 treatments for hypertension (Hall et. al, 1991).

Double blind means neither the patient or the Clinics administering the treatments knew which treatment the patient was receiving.

Treatments:

- A: Carvedilol – new drug being tested for controlling hypertension
- B: Nifedipine – standard drug
- C: Atenolol – standard drug

29 Centers – patients randomized in order of entry

288 patients randomized to receive one of the 3 treatments.

Six visits

Visit	Week	
1	0	Eligibility determined: placebo treatment for one week. Baseline measurements made
2	1	Measurements repeated, eligibility confirmed. Patient randomized to receive treatment
3	3	Measurements repeated
4	5	Measurements repeated
5	7	Measurements repeated
6	9	Measurements repeated

30 patients dropped by the end of the study.

Primary endpoint diastolic blood pressure (DBP). Data available at course website (bp.dat).

Modelling the data

H_0 : no differential treatment effect on DBP.

Model A: assess just treatment effects

$$DBP_i = \mu + \tau_k + e_i$$

where DBP_i = DBP at final visit for patient i

μ = intercept

τ_k = kth treatment effect (for treatment patient i received)

Use ‘last value carried forward’ approach. Treatment may influence dropout, so omitting patients who dropped could lead to biased estimates of the treatment effect. Use last value measured as final visit value for patient.

SAS CODE:

```
FILENAME BP 'C:\USERS\KATHY\STATISTICS DEPARTMENT\STATISTICS 892- MIXED
MODELS\DOWNLOADED STUFF\BROWN AND PRESCOTT\BP.DAT';
DATA BPD;
INFILE BP;
INPUT PAT VISIT CENTER TRT $ DBP DBP0 CF CF1;
RUN;

PROC SORT;
BY PAT VISIT;
RUN;
*GET THE LAST RECORD FOR EACH PATIENT FOR THE LAST RECORD CARRIED FORWARD;

DATA LBPD;
SET BPD;
BY PAT;
IF LAST.PAT;
RUN;

PROC MIXED;
CLASS TRT;
MODEL DBP=TRT;
LSMEANS TRT/PDIFF;
RUN;
```

Cov Parm Estimate
 Residual 81.5660

The Mixed Procedure

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
trt	2	285	2.73	0.0670

Least Squares Means

Effect	trt	Estimate	Standard Error	DF	t Value	Pr > t
trt	A	91.6300	0.9031	285	101.46	<.0001
trt	B	90.4194	0.9365	285	96.55	<.0001
trt	C	88.6211	0.9266	285	95.64	<.0001

Differences of Least Squares Means

Effect	trt	_trt	Estimate	Standard Error	DF	t Value	Pr > t
trt	A	B	1.2106	1.3010	285	0.93	0.3529
trt	A	C	3.0089	1.2939	285	2.33	0.0208
trt	B	C	1.7983	1.3174	285	1.36	0.1733

Model B: Including a baseline covariate

Reasonable to assume relationship between pre- and post- treatment blood pressure measures within a patient.

```
PROC SORT DATA=LDBP;
BY TRT;
PROC MEANS;
BY TRT;
VAR DBP0;
RUN;
```

TRT	N	Mean	Std Dev	Minimum	Maximum
A	100	103.040000	5.2124947	95.000000	120.000000
B	93	102.3440860	4.3825113	95.000000	117.000000
C	95	103.1578947	4.7943135	92.000000	115.000000

Model including baseline blood pressure as a covariate.

$$DBP_i = \mu + \tau_k + \beta x_i + e_i$$

where DBP_i = DBP at final visit for patient i

μ = intercept

τ_k = kth treatment effect (for treatment patient i received)

β =baseline covariate effect

x_i = baseline (pre-treatment) DBP

SAS CODE:

```
PROC MIXED;  
CLASS TRT;  
MODEL DBP=TRT DBP0;  
LSMEANS TRT/PDIFF;  
RUN;
```

Cov Parm	Estimate
Residual	79.8201

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
trt	2	284	2.92	0.0558
dbp0	1	284	7.23	0.0076

Note that the residual variance is reduced when the baseline DBP is included in the model.

Model C: Center effects (center treated as fixed)

The study took place in 29 different centers. The following table shows the distribution of patients to centers and treatments.

Center	Treatment			Total
	A	B	C	
1	13	14	12	39
2	3	4	3	10
3	3	3	2	8
4	4	4	4	12
5	4	5	2	11
6	2	1	2	5
7	6	6	6	18
8	2	2	2	6
9	0	0	1	1
11	4	4	4	12
12	4	3	4	11
13	1	1	2	4
14	8	8	8	24
15	4	4	3	11
18	2	2	2	6
23	1	0	2	3
24	0	0	1	1
25	3	2	2	7
26	3	4	3	10
27	0	1	1	2
29	1	0	2	3
30	1	2	2	5
31	12	12	12	36
32	2	1	1	4
35	2	1	1	4
36	9	6	8	23
37	3	1	2	6
40	1	1	0	2
41	2	1	1	4
Total	100	91	94	288

Centers have different staff, different levels of severity of illness of patients entered into the study, use different labs, are in different areas of the country, etc. Very well could be differences in blood pressure due to centers.

$$DBP_i = \mu + \tau_k + \beta x_i + c_j + e_i$$

where DBP_i = DBP at final visit for patient i

μ = intercept

τ_k = k th treatment effect (for treatment patient i received)

β =baseline covariate effect

x_i = baseline (pre-treatment) DBP

c_j = the j th center effect

SAS CODE:

```
PROC MIXED;  
CLASS TRT CENTER;  
MODEL DBP=TRT CENTER DBP0/SOLUTION;  
ESTIMATE 'A-B' TRT 1 -1 0/ E CL;  
ESTIMATE 'A-C' TRT 1 0 -1/ E CL;  
RUN;
```

Cov Parm	Estimate
Residual	71.9213

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
trt	2	256	2.96	0.0535
center	28	256	2.11	0.0013
dbp0	1	256	3.87	0.0501

Centers do account for a significant source of variation, which causes a reduction in the residual error. This smaller error allows the treatment effect to be calculated with greater accuracy.

Solution for Fixed Effects

Effect	trt	center	Estimate	Standard Error	DF	t Value	Pr > t
Intercept			65.5796	12.9401	256	5.07	<.0001
trt	A		2.9907	1.2336	256	2.42	0.0160
trt	B		1.7937	1.2651	256	1.42	0.1574
trt	C		0
center		1	2.0297	4.4980	256	0.45	0.6522
center		2	-4.5780	5.0511	256	-0.91	0.3656
.							
.							
.							
center		41	0
dbp0			0.2230	0.1133	256	1.97	0.0501

Remember that the estimates of the fixed effects (treatment and center) are biased estimates. We can look at linear functions of the estimates using the ESTIMATE statement.

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper
A-B	1.1970	1.2377	256	0.97	0.3344	0.05	-1.2403	3.6343
A-C	2.9907	1.2336	256	2.42	0.0160	0.05	0.5613	5.4200

Model D: Including a center by treatment interaction effect

The response of patients to treatments may vary between centers for some of the same reasons that centers differ. If some centers tended to enroll more severely ill patients, then it is possible that the reaction of these patients to the treatments may be different than for less ill patients enrolled at other centers.

$$DBP_i = \mu + \tau_k + \beta x_i + c_j + (c\tau)_{jk} + e_i$$

where DBP_i = DBP at final visit for patient i

μ = intercept

τ_k = kth treatment effect (for treatment patient i received)

β =baseline covariate effect

x_i = baseline (pre-treatment) DBP

c_j = the jth center effect

$(c\tau)_{jk}$ = the kth treatment effect at the jth center

This model assumes that treatment effects are specific to the center observed. This is a *hierarchical* model because treatment effects are *contained* within the center*treatment effects. When the effect of interest (in our case treatment) is contained within another fixed interaction effect (treatment*center), the coefficients for the interaction effect are automatically included in the estimate by SAS. Therefore, only the coefficients for TRT need to be indicated in the estimate statement.

```
PROC MIXED;
CLASS TRT CENTER;
MODEL DBP=TRT|CENTER DBP0/ SOLUTION;
ESTIMATE 'A-B' TRT 1 -1 0/ E CL;
ESTIMATE 'A-C' TRT 1 0 -1/ E CL;
```

Effect	Coefficients for A-B		
	trt	center	Row1
Intercept			0
trt	A		1
trt	B		-1
trt	C		0
center		1	0
center		2	0
.			
trt*center	A	1	0.0385 = 1/n _a where n _a is the number
trt*center	A	2	0.0385 of centers where treatment A
trt*center	A	3	0.0385 was received (26).
.			
trt*center	B	1	-0.04 = 1/n _b where n _b is the number of
trt*center	B	2	-0.04 centers where treatment B
.			
trt*center	C	1	0
trt*center	C	2	0

Cov Parm	Estimate
Residual	69.2614

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
trt	2	208	1.24	0.2905
center	28	208	1.98	0.0038
trt*center	48	208	1.20	0.1884
dbp0	1	208	0.99	0.3198

The treatment*center effect is not significant.

Solution for Fixed Effects

Effect	trt	center	Estimate	Standard Error	DF	t Value	Pr > t
Intercept			76.7901	15.6445	208	4.91	<.0001
trt	A		-2.3811	10.1999	208	-0.23	0.8157
trt	B		8.9839	11.8136	208	0.76	0.4478
trt	C		0
center		1	-1.5370	8.6673	208	-0.18	0.8594
center		2	0.4234	9.6192	208	0.04	0.9649
.							
center		40	0.03230	11.9447	208	0.00	0.9978
center		41	0
trt*center	A	1	6.1891	10.7253	208	0.58	0.5645
trt*center	A	2	-3.2910	12.2502	208	-0.27	0.7885
.							
trt*center	C	36	0
trt*center	C	37	0
trt*center	C	41	0
dbp0			0.1270	0.1274	208	1.00	0.3198

Because treatments A, B and C were not received at every center, the estimate is ‘non-estimable’.

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper
A-B	Non-est
A-C	Non-est

Model E: Including Center and Center*Treatment effects as random

The centers that participated in the study are a sample of centers from a much larger population of centers that would have qualified to participate in the study. Therefore, we might want to consider the center and therefore the center*treatment effects as arising from a normal distribution.

$$DBP_i = \mu + \tau_k + \beta x_i + c_j + (c\tau)_{jk} + e_i$$

where DBP_i = DBP at final visit for patient i

μ = intercept

τ_k = k th treatment effect (for treatment patient i received)

β = baseline covariate effect

x_i = baseline (pre-treatment) DBP

c_j = the j th center effect

$(c\tau)_{jk}$ = the k th treatment effect at the j th center

$$e_i \sim N(0, \sigma^2)$$

$$c_j \sim N(0, \sigma_c^2)$$

$$(c\tau)_{jk} \sim N(0, \sigma_{ct}^2)$$

Because we assume that differences between treatments vary randomly across the centers, we can relate the results to the population of potential centers.

ANOVA table

Source of Variation	E(M.S.)
Treatment	$\sigma^2 + 3\sigma_{ct}^2 + (3*29)\Sigma\tau_i^2$
Center	$\sigma^2 + (3*3)\sigma_c^2$
Center*Treatment	$\sigma^2 + 3\sigma_{ct}^2$
Residual	σ^2

```
PROC MIXED;
CLASS TRT CENTER;
MODEL DBP=TRT DBP0;
RANDOM CENTER TRT*CENTER;
ESTIMATE 'A-B' TRT 1 -1 0/ E CL;
ESTIMATE 'A-C' TRT 1 0 -1/ E CL;
RUN;
```

Cov Parm	Estimate
center	6.4628
trt*center	4.0962
Residual	68.3677

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
trt	2	48	2.28	0.1131
dbp0	1	208	6.31	0.0128

Coefficients for A-B

Effect	trt	Row1
Intercept		
trt	A	1
trt	B	-1
trt	C	0
dbp0		0

Coefficients for A-C

Effect	trt	Row1
Intercept		
trt	A	1
trt	B	0
trt	C	-1
dbp0		0

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper
A-B	1.2859	1.3914	48	0.92	0.3600	0.05	-1.5116	4.0834
A-C	2.9274	1.3727	48	2.13	0.0381	0.05	0.1674	5.6874

Following are the estimates for the same two estimates from Model C, where center was treated as a fixed effect.

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper
A-B	1.1970	1.2377	256	0.97	0.3344	0.05	-1.2403	3.6343

A-C 2.9907 1.2336 256 2.42 0.0160 0.05 0.5613 5.4200

Note that the standard errors of the treatment estimates from the random center model are inflated when compared to the standard errors from the fixed center model. Should the center and the center*treatment effects be treated as fixed or random effects? There are no hard and fast rules. Various approaches will be covered in detail later in the course.

Treating center and center*treatment effects as random allows the inferences that we make to apply to the whole population of centers, while treating center as fixed (and dropping out the non-statistically-significant center*treatment term) allows for more precise treatment estimates.

Repeated Measures Data

We have been analyzing only the last record for each patient in the multi-center trial. However, there were up to four post-treatment visits for each patient. *Repeated measures data* is when measures are made repeatedly on the same patient.

Covariance pattern models

The primary objective is to assess the effect of the treatments on DBP, but now we want to consider all observations per patient. There could also be a possible underlying change in DBP over the four post-randomization visits which can be allowed for by including a time effect (m). A treatment*time interaction can also be included to account for differential treatment effect over time.

$$DBP_{ij} = \mu + \tau_k + \beta x_i + m_j + (\tau m)_{jk} + e_{ij}$$

where DBP_{ij} = DBP at the j th post-treatment visit for patient i

μ = intercept

τ_k = k th treatment effect (for treatment patient i received)

β = baseline covariate effect

x_i = baseline (pre-treatment) DBP

m_j = time effect at the j th post-treatment visit

$(\tau m)_{jk}$ = the k th treatment effect at the j th post-treatment visit

We need to also take into account that post-treatment measures taken on the same patient are not independent.

$$\text{corr}(e_{ij}, e_{ij'}) = \rho, j \neq j'$$

$$\text{corr}(e_{ij}, e_{ij'}) = \rho^{|j'-j|}, j \neq j'$$

$$\text{corr}(e_{ij}, e_{ij'}) = \rho_{j-j'}, j \neq j'$$

These are three different types of covariance patterns that can be used to fit the covariances or correlations. Fitting covariance patterns with repeated measures data

- lead to more appropriate analysis
- covariance parameter estimates may also uncover additional information about the data.

Random coefficients models

An alternative approach to modelling repeated measures data.

Explain arithmetically the relationship between the dependent variable and time.

Simple way – include a quantitative time effect as a covariate in the model.

$$\text{DBP}_{ij} = \mu + \tau_k + \beta x_i + m * \text{time}_{ij} + e_{ij}$$

where DBP_{ij} = DBP at the j th post-treatment visit for patient i

μ = intercept

τ_k = k th treatment effect (for treatment patient i received)

β = baseline covariate effect

x_i = baseline (pre-treatment) DBP

m = time covariate effect

time_{ij} = time of observation j for patient i (weeks)

Obtain a time slope with gradient m which defines a linear relationship between DBP and visit.

However, this relationship could vary between patients. Allow for this by fitting a separate regression for each patient.

$$DBP_{ij} = \mu + \tau_k + \beta x_i + p_i + m * time_{ij} + (pm)_i * time_{ij} + e_{ij}$$

where DBP_{ij} = DBP at the j th post-treatment visit for patient i

μ = intercept

τ_k = k th treatment effect (for treatment patient i received)

β = baseline covariate effect

x_i = baseline (pre-treatment) DBP

p_i = difference from average in the intercept term for the i th patient

$(pm)_i$ = difference in slope for the i th patient, from the average slope

m = time covariate effect

$time_{ij}$ = time of observation j for patient i (weeks)

Reasonable to assume that patients and their slopes come from a distribution and should be fitted as random effects. The statistical properties of a model where some of the random effects involve covariate terms are different from ordinary random effects models and are referred to as *random coefficients models*.

Distributional assumptions for a random coefficient model.

$$\text{var}(e_{ij}) = \sigma^2$$

With a random coefficient model, the estimates of the slope and the intercept are not independent.

The patient effects (intercept) and patient*time effects (slope) are correlated within patient.

Use a bivariate normal distribution:

$$\begin{pmatrix} p_i \\ pm_i \end{pmatrix} \sim \mathbf{N}(\mathbf{0}, \mathbf{G})$$

where

$$\mathbf{G} = \begin{pmatrix} \sigma_p^2 & \sigma_{p,pm} \\ \sigma_{p,pm} & \sigma_{pm}^2 \end{pmatrix}$$

Repeated measures can be modelled either with the covariance pattern model or the random coefficients model. The random coefficient model is usually more appropriate:

- 1) if the repeated measures do not occur at fixed intervals
- 2) the relationship with time is of particular interest

Mixed Models

Mixed models compared with fixed effects models are able to model data in which the observations are not independent. Mixed models are able to model the covariance structure of the data.

Why use mixed models?

- Fitting covariance pattern models leads to more appropriate fixed effect estimates and standard errors
- Results from a mixed model may be more appropriate to the required inference when the data structure is hierarchical.
- In a cross-over trial estimates of treatment effects can become more accurate in datasets where there are missing data.
- In a random effects model estimates of random effects are ‘shrunk’ compared with their fixed effects counterparts.
- Different variances can be fitted in a mixed model for each treatment group.
- Problems caused by missing data when fitting fixed effects models do not arise in mixed models.

Potential disadvantages

- distributional assumptions are made
- approximations are used to estimate certain model parameters
- Therefore conclusions are dependent on more assumptions being valid and will be some circumstances where parameters are biased.